USING IMPORT TO IMPLEMENT COMPLEX BEHAVIORS IN SIMULATIONS





Jeffrey W.  Wallace


Naval Command, Control and Ocean Surveillance Center, RDT&E Division, Code D44202


San Diego, California





R. Alan Whitehurst


Department of Computer Science


Brigham Young University


Provo, Utah











�
ABSTRACT


The Declarative Object Management Environment (DOME) is the declarative sublanguage of the Integrated Modular Persistent Objects and Relations Technology (IMPORT) computer language [Whitehurst 95].  DOME implements a logic programming language similar to PROLOG [Sterling 91] which allows the description of complex behaviors.  DOME can also be used to create ìshallow modelsî where the programmer needs only a very rough model of some very complex object (for example modeling the economy based on various statistical indexes).  While in theory logic programs can do any computation, in practice they are inefficient at performing arithmetic and most algorithmic computations. An optimal combination of both paradigms is achieved by combining an imperative and declarative language.  





INTRODUCTION


Over the last two decades, researchers have attempted to develop large scale simulations spanning many levels of system resolution.  The wargaming models developed by operations researchers in support of defense department decision makers are an example. These efforts to develop simulations of large parts of reality, from switching circuits to human decision making, have made use of most of what is known about software technologies and techniques.


For example, declarative programming is typically used to model human decision making, while static portions of models are written in imperative languages for efficiency.  However, most expert systems are designed as stand-alone packages and provide only loosely coupled interfaces to imperative languages.  This is true of other software technologies and has hampered their general use to address modeling and simulation in a consistent and integrated manner.  Recent Defense Advanced Research Projects Agency (DARPA) and Department of Defense (DoD) initiatives have goals far beyond the use of these large simulations as single-user analytical tools.  They have embarked on ambitious efforts at providing distributed interactive simulations using virtual reality for training, virtual prototyping of hardware, and analytical applications. 


The Integrated Simulation Language Environment (ISLE) is a project of the Collaborative, Object-oriented Advanced Simulation Technology Research (COASTeR) project at the Naval Command, Control and Ocean Surveillance Center, Research, Development, Test & Evaluation Division (NRaD).  The goal of ISLE is to identify and develop enabling technology for intelligent, collaborative simulation-based software engineering environments.  The domain of interest is the construction of large-scale, complex simulations. 


The ISLE project includes compiler and interpreter development for the IMPORT fifth-generation computer language (5GL) for computer simulation that has a declarative sublanguage, DOME, for modeling complex behaviors.  This paper provides a brief overview of IMPORT, describes DOME, and then illustrates its use through examples.





IMPORT OVERVIEW


IMPORT [Whitehurst 95, NCCOSC 97] is a 5GL tailored to develop large-scale, complex simulations.  It is the latest development in a chain of languages that can be traced through MODULA-2 [Wirth 84], Ada [Cohen 86], MODSIM [CACI 89], and MODLOG [Whitehurst 91].


IMPORT is fashioned after modern, structured-language principles, such as the third-generation languages C++ and Ada95.  This includes object-oriented modeling, inheritance, encapsulation, and polymorphism.  IMPORT also supports information hiding and data abstraction through module structures with well-defined interfaces.


IMPORT advances beyond the third generation languages and provides language constructs for  time management, which allows object-level concurrency, along with synchronization mechanisms to allow arbitrarily complex logical synchronization between sets of objects.  The following code fragment, a server class from a queueing network simulation framework, provides an example of IMPORT syntax:





MODULE Server;





IMPORT RANDOM;


IMPORT TASKS;


IMPORT STRINGS;


IMPORT Token;


IMPORT SimQueue;


IMPORT System;


IMPORT Collectable;





TYPE





ServerObj = OBJECT(Collectable)





  RV : Random;


  ID : INTEGER;


  Status : REAL;


  queue : SimQueueObj;


  secondaryQueue : SimQueueObj;


  secondaryQueuePresent : BOOLEAN;


  System : SystemObj;


  ServiceTime : INTEGER;


	


CONSTRUCTOR METHOD ServerObj()


INIT Collectable()


BEGIN


  Status := 0.0;


  secondaryQueuePresent := FALSE;


END METHOD;





Ö





TELL METHOD ProcessToken(IN token : Token)


VAR


  newOne : TokenObj;


  delay : INTEGER;


BEGIN


  MakeBusy();


  WAIT(ServiceTime);


  ASK System TO UpdateStatistics(SIMTIME());


  ASK System TO RouteToken(token);





  IF ASK queue IsEmpty()


    IF secondaryQueuePresent


      IF ASK secondaryQueue IsEmpty()


        MakeIdle();


      ELSE


        newOne := ASK secondaryQueue TO Remove();


        delay := SIMTIME() - ASK newOne TO GetArrivalTime();


        ASK secondaryQueue TO AddToTotalDelay(delay);


        ProcessToken(newOne);


      END IF;


    ELSE 


      MakeIdle();


    END IF;


  ELSE


    newOne := ASK queue Remove();


    delay := SIMTIME() - ASK newOne ArrivalTime();


    ASK queue TO AddToTotalDelay(delay);


    ProcessToken(newOne);


  END IF;


END METHOD;





END OBJECT;





END MODULE.





Researchers and developers familiar with MODSIM or MODSIM II and III will note the similarities in syntax.  However, the implementation and features of IMPORT are quite different [Whitehurst 95].  The sequel describes just one of the major differences between IMPORT and these languages, namely the inclusion of a declarative sublanguage to model complex behaviors.





LOGIC PROGRAMMING


A good way to think about a logic program is simply as a database of facts.  A logic program executes by performing a query.  Queries can ask if a particular fact is in the database, or more commonly, if there is any object in the database that satisfies a particular condition.  Executing the query returns a Boolean result indicating whether or not the fact was in the database, and in the case of a successful result, the fact that made the query succeed.  This second part of the answer is important because it is often more important than the Boolean result.  For example, a database of scheduling information might be queried for a free time slot.  Knowing that the answer is yes is not very useful unless the query also returns additional information about the free time slot.  


What makes logic programming powerful, is the way the database of facts can be specified.  In addition to just specifying facts, inference rules can also be specified.  The database can be thought of as having not only the explicitly specified facts, but also all those facts that can be derived from the inference rules.  Since inference rules can be recursive (they can generate a new fact to which the same inference rule can be applied recursively), the logical database of facts is often quite large (usually infinite).  This is what makes logic programming powerful.  


Unfortunately, this very power leads to a drawback.  Recursive inference rules can make the interpreter for the logic program run forever when it is trying to answer a query.  Exactly when this occurs depends on both the inference rules and the query.  Thus, to understand termination behavior, this simple model of a logic program as a virtual database of facts has to be replaced by a more algorithmic description of what the logic program interpreter does.  This will be described later.  For now, the simple model of a logic program will suffice.  





TERMS AND CLAUSES


Compared to IMPORT, the syntax of DOME is quite simple.  There are only two major syntactic components, Terms and Clauses.  The basic syntax is listed here.  





 	Clause		: <Term>.            


		| IF <Term1> AND ...  AND <TermN> THEN <Term>.            





	Term		: <Id>(<Term1>, ..., <TermN>)


		| ?<Id>


		| [<Term1>, ..., <TermN>]


		| [<Term1>, ..., <TermN> | <TermL> ]


		| <| <Exp> |>


		| <<| <Stmts> |>>


		| <Literal>


	;





Terms


There are seven kinds of terms used to represent objects in the database:  


<Id>(<Term1>, ..., <TermN>) 


Functors.  Functors are the central kind of term in a logic programming language.  Most facts in a database are formed from these.  If a functor has no arguments, the parentheses are not needed.  Functors are used both to describe predicates as well as objects on which the predicates operate.  For example the fact:


Distance(Champaign, Chicago, 140)


might be used to state that the cities of Champaign and Chicago are 140 miles apart.  In this case, Distance is a functor used as a predicate, and Champaign and Chicago are functors used to represent objects in the world.  


?<Id> 


Variables.  Variables are differentiated from functors with no arguments by the prefix `?` operator.  Variables represent arbitrary facts in the database as detailed below.  


[<Term1>, ..., <TermN>] 


Literal Lists.  This form represents a list of the given terms.  


[<Term1>, ..., <TermN> | <TermL> ] 


Prepending Lists.  This form represents the list <TermL> with the terms <Term1>, ..., <TermN> inserted in the beginning.  


<| <Exp> |> 


IMPORT Constants.  Any IMPORT constant can be an object in the DOME database using the form.  <Exp> must be a constant expression (often a value of an enumerated type).  


<<| <Stmts> |>> 


IMPORT Call.  This is the mechanism by which DOME can ask IMPORT to perform calculations that would be difficult or impossible for DOME to do.  It is described in detail later.  


<Literal> 


Any literal constant can be put in the DOME database.  This form is really just shorthand because it could be expressed using the IMPORT Constant form (e.g.,  34 is the same as <| 34 |>).  


Clauses


Clauses represent facts in the logic database.  As stated earlier there are two kinds of data in the database, simple facts and inference rules.  The first clause form can be used to assert a simple fact, for example the clause:


Distance(Champaign, Chicago, 140).


might be used to assert the fact that the cities of Champaign and Chicago are 140 miles apart.  


Inference rules are stated using the second type of clause.  Typically, Inference rules use variables to represent arbitrary DOME terms.  For example the clause: 


IF Distance(?from, ?from, 0) 


THEN Reachable(?from, ?to).


asserts that whenever there is a fact in the database specifying the distance between two places, then it is also true that the two places are reachable from one another.  


More generally, an inference rule asserts that whenever facts in the database can make the IF part of the inference rule true by a suitable binding of the variables in the clause, the THEN part of the clause is also in the database (instantiated with the same binding).  This implies that the scope of variables extends only over the clause in which they are found.  Other clauses can reuse the same variable names, but logically they are distinct from variables in other clauses.  





QUERY EXECUTION


While the simple database model is a useful way of thinking about DOME programs, a more complicated model of DOME execution is necessary to understand termination behavior and how to optimize DOME programs.  


When the DOME interpreter is activated, it has two basic pieces of information:++-+-+--+  


A term representing the query to search the knowledge-base.  The query term may have variables in it.  These variables have an existential interpretation.  For example the query: ìDistance(?x, ?y, 140)î asks if terms exist for ?x and ?y such that the query is in the database.  (The intention of this query is to ask the question: ìAre there any cities that are 140 miles apart?î) 


The database of clauses to search over.  


The goal of the interpreter is to find a binding for the variables in the query such that the query is a logical deduction of the facts and inference rules in the database.  The interpreter attempts to do this by the following process.  


It examines the first clause in the database and finds its head.  The head of a simple fact is the fact itself.  The head of an inference rule is the term in the THEN part of the clause.  The interpreter uses the process of unification to determine if the head could possibly match the query.  If it cannot, the clause fails and the interpreter tries the next clause.  If the query matches the head of a simple fact, the interpreter can return.  In the case of an inference rule, the interpreter needs to first verify that all the conditions in the IF part of the clause are satisfied.  To do this, the interpreter calls itself recursively on each term in the IF part of the clause.  If any of these subqueries fail, then the clause fails and the interpreter moves on to the next clause as before.  If all the subqueries succeed, however, the interpreter can return success and the binding resulting from the subqueries.  


This algorithm has several consequences: 


It may be possible to satisfy a query in more than one way.  In this case only the first one found using the above algorithm is returned.  


The process is not guaranteed to terminate.  


For example the inference rule:


ìIF Distance(?x, ?y, ?d) THEN Distance(?y, ?x, ?d).î


which is a very reasonable rule from a logical perspective (stating that ìDistanceî is symmetric) and will usually cause an infinite loop.  In particular, the query:


ìDistance(Champaign, ?y, ?d)î


will unify with the inference rule and produce the subquery: 


ìDistance(?y, Champaign, ?d)î


which will in turn produce the original query ìDistance(?y,Champaign, ?d)î.  The interpreter does not detect that it is back where it started and continues the process, creating an infinite loop.  





LINKS TO IMPORT’S IMPERATIVE SUBLANGUAGE





Calling DOME from IMPORT (imperative sublanguage)


Every object in IMPORT can have a DOME database attached to it.  Objects can also have QUERY methods, each of which starts a DOME interpreter on the database.  For example the object USA could be defined as follows:





USA = OBJECT 


   CITIES = (Champaign, Chicago, Miami);


   QUERY METHOD Distance(IN from, to:CITIES; 


                               OUT d:INTEGER):BOOLEAN;


   DOME BEGIN


      Distance(<| Champaign |>, <| Chicago |>, 140).


      Distance(<| Champaign |>, <| Miami |>, 2000).


   END DOME;


END OBJECT;


Invoking the query method of an object starts the DOME interpreter.  The query that the interpreter searches for is formed according to the following rules.  





The query is a single predicate (functor term) with the same name as the query method itself.  





All IN parameters of the query method are passed unchanged as the corresponding argument of the predicate. 





All OUT parameters of the query method are replaced by DOME variables in the corresponding argument of the predicate.  





INOUT parameters are illegal.  


The query method returns the status of the DOME interpreter.  If the query was successful, the value of all OUT parameters are copied back to the actual argument.  If on return the values passed back from DOME do not match the type of the OUT parameter, the QUERY method returns failure.


In the above example the expression 





QUERY SELF ABOUT Distance(Champaign, Chicago, dist);





would return the value TRUE and as a side effect set dist to 140.  


In the above example, it was important that Champaign and Chicago were not ordinary DOME functors but rather IMPORT constants (by surrounding them with <| |>).  IMPORT has no way of representing DOME functors, so there would be no way of retrieving the values of a DOME query that returns DOME functors.  Instead, constants that can be expressed in IMPORT (in this case, an enumerated type) are used instead of functors.  Note that DOME Lists are the same as IMPORT Lists, so lists can be passed between the two languages without a problem.  








Calling IMPORT (imperative sublanguage) from DOME


While DOME excels at performing inferences and binding variables, traditional algorithmic processes, such as arithmetic, are difficult. To resolve this difficulty, DOME has the <<| |>> term that allows it to call IMPORT functions.  In general the <<| |>> brackets enclose a list of statements that make up the body of a Boolean valued method.  These statements are executed every time DOME tries to prove the <<| |>> term.  The following example defines the predicate EVEN by calling IMPORT.  A query involving this predicate, like EVEN(6), would match the above inference rule, causing the interpreter to try to prove the <<| |>> term.  This causes the execution of the IMPORT code inside the <<| |>> brackets.





DOME BEGIN


    IF <<| RETURN((?num ASA INTEGER) MOD 2 = 0); |>>


   THEN EVEN(?num).


END DOME;


The code inside the <<| |>> brackets is not pure IMPORT code.  In particular, DOME variables may be used in this code.  DOME variables are treated as expressions of type TAGGED.  The ISA operator on TAGGED types works as always, with one extension.  A special type called VAR may be used with the ISA operator to determine if the TAGGED cell contains an uninstantiated DOME variable.  Values may be passed back to DOME by assigning a value to such uninstantiated values.  Note, that if assignments are made in IMPORT to DOME variables, that have already been instantiated by DOME, the changes made in IMPORT are ignored.  The IMPORT code should return a Boolean value.  False indicates failure.  On success, the IMPORT code should instantiate any uninstantiated DOME variables to the values that make the predicate true (this is the primary mechanism for passing values back to DOME).


A special variable called INDEX can be used in the IMPORT code.  In general there may be many possible bindings of variables that will satisfy a predicate.  Each of these possibilities must be tested against the other conditions of the query until a complete match is found.  Thus, a predicate must be able to return not just a single binding, but also a list of all possible bindings that satisfy the predicate.  The INDEX variable indicates which of the possibilities DOME is interested in retrieving.  Initially DOME sets INDEX to 0.  If that possibility does not satisfy the other conditions in the query, it increments the index and calls the IMPORT predicate again until either the additional conditions are satisfied, or the IMPORT predicate fails.  


In the EVEN example, the query EVEN(6) would cause ?num to be bound to a TAGGED value containing 6.  The IMPORT code then uses the ASA operator to untag the number, test it with the MOD function, and return the result (true in this case).  


Note, that while this code may be sufficient for some purposes, it is not robust enough for general purpose use.  One problem is that DOME might be given the query EVEN("hello") which would cause ?num to be bound to a TAGGED value containing a string.  The ASA operator will cause a run time error in this case.  A more serious problem is that the query EVEN(?x) would also cause a run time error.  This is more serious because the query EVEN(?x) should succeed and bind ?x to an even number.  Code that fixes these problems is shown below.  





DOME BEGIN


   IF <<| IF (?num ISA VAR) THEN


      ?num := INDEX * 2;


      RETURN(TRUE);


   ELSEIF (?num ISA INTEGER) THEN  


      RETURN((?num ASA INTEGER) MOD 2 = 0); 


   ELSE 


      RETURN(FALSE); 


   END IF; |>>


   THEN EVEN(?num).


END DOME;


Here the ISA operator is used to test what the variable ?num is.  If it is a variable, the INDEX variable is used to bind ?num to the appropriate even number.  If ?num is an instantiated INTEGER, it is tested.  If ?num is not an INTEGER, the predicate fails.  





BUILT IN PREDICATES


Just as a run time library is useful in IMPORT, a set of predefined predicates can save effort in DOME.  In theory most of these predicates are defined for convenience since they could have been defined using the operators already described.  A few of them, however, manipulate the database in ways that could not otherwise be done.  A complete list of all DOME primitives is given in [NCCOSC 97].  





Convenience Primitives


The Convenience Primitives can be divided into five groups: arithmetic, comparison, type probing, string related, and miscellaneous.  For the most part, these functions are simple and standard enough that even the very terse descriptions in [NCCOSC 97] are sufficient to use them properly.  In general operators work on whatever values make sense, arithmetic on integers and reals, or any other valid variable type.  For example





plus(5, 3, ?x)





would set the variable ?x to 8.  Note that when it is feasible primitives will succeed when used “backwards”.  For example





plus(5, ?x, 8)





would set the variable ?x to 3.  When inverting the function is infeasible or when there are just too many possibilities, the primitive will fail instead.  For example, both of these predicates would fail if the variables are unbound.  





plus(?y, ?x, 8)


plus(?y, ?x, ?z)


Strictly speaking, returning FALSE is not correct for predicates like this since it is possible to satisfy the predicate.  In practice, any program that tried to use terms like the ones above would either not terminate or take an unreasonable amount of time to complete.  Thus, the fact that the primitives simply fail is not usually a problem.  





Meta Primitives


The following DOME predicates are really more a part of the programming environment than of the language.  They allow a programmer to probe and influence the DOME interpreter in various ways that ordinary predicates simply cannot do.  


[trace] 


Toggles debug tracing on or off.  This predicate is the primary way of debugging a DOME program that is currently available.  It allows the user to watch each step of the attempt to answer a query.  It can be too verbose at times, but it is highly useful nevertheless.


[prims] 


This primitive prints a list of primitive names.  In theory, this list should be the same as the list of primitives [NCCOSC 97].  In practice, it is likely to contain extra, newly implemented, or experimental primitives.  





EXAMPLES


We now describe two examples that were developed using the approach described above.  Space does not permit the source code for these applications to be presented, but the content of their knowledge-bases and simulation functionality will be presented.





Flexible Manufacturing Simulation


This example is from an analysis of a wood molding manufacturing line, performed at the University of Illinois’ Department of Industrial Engineering.  This example demonstrates two important principles.  First, this illustrates how simulations can be constructed that allow strategy and decision making to be implemented in a modular fashion so that several strategies can be tested without altering the simulation; one merely substitutes different knowledge-bases.  Second, this demonstrates the importance of factoring human behavior into analyses to fully model situations where humans are involved.


The wood molding manufacturing line has three operations: drilling, routing, and shaping using lathes. Each machine also has two input queues: one designated as the primary queue, and the other as the secondary queue,  The machines can be quickly reconfigured to change these designations.  In addition, a flexible machine that can be used as a drill or a router is also available.  Changing the function requires stopping the machine and performing a short re-tooling.  It requires a person to perform the re-tooling operation, and this person may not always be immediately available.  The factory supervisor makes the decisions about queue designations, whether or not the flexible machines are used, and which operation they should perform.


The scenario is configured as follows:





	Two part types, labeled A and B


	Two drills


	Two lathes


	Two routers


	One flexible machine (router or drill)


	One factory supervisor (the decision maker).


Part A follows the path: router-lathe-drill.  Part B follows the path: drill-router-lathe.  The supervisor periodically examines the state of the work-in-progress queues and decides whether or not to employ the flexible machine and whether (and how) to reconfigure the queue priorities.  We note that work-in-progress parts accumulate in one queue, and new parts arriving to the system accumulate in the other.  Since new parts enter the system only through the drills and routers, one would expect conflicts between work-in-progress and new arrivals in this area of the manufacturing line.


Three knowledge-bases illustrate how this approach could be used to determine the optimal policy for flexible machine usage and queue management.  We emphasize that only the knowledge-bases change ( none of the simulation code changes.  Different decisions are made depending on which knowledge-bases are used, leading to the simulation of different actions.


The first knowledge-base is the control.  Thus, the flexible machine is not used, and the new part arrival queues are given priority over the work-in-progress queues for the drills and routers.  The average makespan for a part in this configuration is 2299.57 seconds.  Analysis of the machine utilization shows that the drills and routers are bottlenecks, since both must service both new part arrivals and work-in-progress arrivals.


The second knowledge-base examines the arrival queues for new part arrivals to the system for both drills and routers.  If the drill arrival queue is larger when the supervisor inspects the system, the flexible machine is configured as a drill, and it draws the next part from the new part arrival queue for the drills.  If the situation is reversed, the flexible machine is configured as a router and parts are drawn from the new part arrival queue for the routers.  The average makespan is now 127.57 seconds, an improvement by a factor of more than 18 over the control scenario.


The third knowledge-base is more complex, and takes into account not just the new part arrival queues but also the state of the work-in-progress queues.  Now when the supervisor inspects the system, the decision can be made to not just configure the flexible machine as a drill or router, but to determine whether the flexible machine should draw from the new part arrival queue or the work-in-progress queue.  The average makespan for this scenario is 89.48 seconds, a further improvement by a factor of over 1.4 from the previous scenario.





Wargaming and Training Simulation


Wargaming and training simulation in the military is more crucial than ever because of fiscal constraints and the increasing quality of military hardware available to potential adversaries.  How capable we are in deploying and employing our assets is of paramount importance.  In [Wallace 91] a prototype wargaming and training system was implemented in the predecessor to IMPORT, the MODSIM/MODLOG language system.  The key architectural feature of this prototype is the natural representation of the military command hierarchy in an object-oriented framework which factored out command behavior explicitly.  This was achieved through the pairing of a Commander base class with a Military Unit base class.  


The Commander class and its descendants, utilized the declarative sublanguage MODLOG, from which DOME evolved, to capture complex, high-level decision making.  The Military Unit class and its descendants then executed the wishes of the associated Commander objects.  The usage of IMPORT, in an updated version of this architecture, is currently being explored for use in the next generation of wargaming and training simulations.





CONCLUSIONS


The use of fifth generation languages to solve increasingly complex problems is now being widely recognized.  As resources become more scarce, the use of simulation to solve complex analysis and training problems will be more important, and the IMPORT 5GL for simulation has been demonstrated to accommodate these problems.  Exploring the virtues of fully integrated declarative and imperative sublanguages is just beginning, and the use of rules in AI techniques such as neural networks, evolutionary programming, and genetic algorithms will lead to even more powerful representations of cognitive functions.

